


[← How to Create Waterfall Charts in Excel](#)[Extract a List of Values Filtered by Criteria with Sub-Arrays](#)

→

How to VLOOKUP with Multiple Criteria Using INDEX and MATCH

BY ANDREW ROBERTS



VLOOKUP is a great tool for pulling data from tables, but it has a handicap: it can only work with one criteria for matching information. If there are multiple rows in your sheet with the same information, you'll only get the first one. If you need to use two or more conditions to match a specific piece of data, you're out of luck. Fortunately, Excel has a pair of functions called **INDEX** and **MATCH** that can help produce the same

results as **VLOOKUP** with multiple criteria. Here's a quick tutorial to help you learn how...

Example Data

Let's say, for example, that we want to be able to search through a list of fuel economy data for cars to find the mileage...

Quick Navigation [hide]

- [1 Example Data](#)
- [2 Trying to Use VLOOKUP](#)
- [3 Using INDEX and MATCH to Replace VLOOKUP](#)
- [4 Using INDEX and MATCH with Two Criteria](#)
- [5 Multiple Criteria VLOOKUP with INDEX and MATCH Example Download](#)
- [6 Other posts in this series...](#)
- [6.1 Share this:](#)

	A	B	C	D	E	F	G	H	I
1	Year	Make	Model	Disp	# Cyl	City	Hwy	Comb	En
2	2014	Acura	ILX	1.5	4	39	38	38	Na
3	2014	Acura	ILX	2.0	4	24	35	28	Na
4	2014	Acura	ILX	2.4	4	22	31	25	Na
5	2014	Acura	MDX 2WD	3.5	6	20	28	23	Na
6	2014	Acura	MDX 4WD	3.5	6	18	27	21	Na
7	2014	Acura	RDX 2WD	3.5	6	20	28	23	Na
8	2014	Acura	RDX 4WD	3.5	6	19	27	22	Na
9	2014	Acura	RLX	3.5	6	20	31	24	Na
10	2014	Acura	RLX	3.5	6	28	32	30	Na
11	2014	Acura	TL 2WD	3.5	6	20	29	23	Na
12	2014	Acura	TL 4WD	3.7	6	18	26	21	Na

Normally, we would want to be able to enter the model of a car and get it's fuel economy as a result. Unfortunately, Many cars, like the Acura ILX, have multiple engine configurations with different mileage ratings. Fortunately, in this case, the car's displacement can serve to separate them.

	A	B	C	D	E	F	G	H	I
1	Year	Make	Model	Disp	# Cyl	City	Hwy	Comb	En
2	2014	Acura	ILX	1.5	4	39	38	38	Na
3	2014	Acura	ILX	2.0	4	24	35	28	Na
4	2014	Acura	ILX	2.4	4	22	31	25	Na
5	2014	Acura	MDX 2WD	3.5	6	20	28	23	Na
6	2014	Acura	MDX 4WD	3.5	6	18	27	21	Na

This means, however, that we will need to look up the car by both its **Model** and its **Displacement** at the same time to find the appropriate **Combined Fuel Economy** in column H.

Trying to Use VLOOKUP

In a normal **VLOOKUP**, the syntax is as follows:

```
=VLOOKUP(lookup_value, table_array, col_index_num, [range_lookup])
```

The *lookup_value* is the data you are searching with.

The *table_array* defines the table that you want to look through. The first column must be the column that has the *lookup_value* in it.

The *col_index_num* is the number of the column in the *table_array* that has the data you want to find.

The optional *range_lookup* specifies whether the list is sorted or not. (**TRUE** means that **VLOOKUP** stops looking when it finds something that comes later in the alphabet than the *lookup_value*. **FALSE** means it searches the entire list.)

If we were looking for just the Model of car using **VLOOKUP** in our example data, it would look like this:

	A	B	C	D	E	F	G	H
1	Model	Combined Economy						
2	ILX	=VLOOKUP(A2,C4:H1159,6,FALSE)						
3								
4	Year	Make	Model	Disp	# Cyl	City	Hwy	Comb
5	2014	Acura	ILX	1.5	4	39	38	38
6	2014	Acura	ILX	2.0	4	24	35	28
7	2014	Acura	ILX	2.4	4	22	31	25
8	2014	Acura	MDX 2WD	3.5	6	20	28	23
9	2014	Acura	MDX 4WD	3.5	6	18	27	21
10	2014	Acura	RDX 2WD	3.5	6	20	28	23
11	2014	Acura	RDX 4WD	3.5	6	19	27	22
12	2014	Acura	RLX	3.5	6	20	31	24

```
=VLOOKUP (A2, C4:H1159, 6, FALSE)
```

A2 holds the Model of car we want to find.

C4:H1159 is the table we want to search through. Column C is the column with the **Model** information.

Column **6** is the column that holds the **Combined Fuel Economy** figure that we want to find.

The result of the **VLOOKUP** is this:

	A	B	C	D	E	F	G	H
1	Model	Combined Economy						
2	ILX		38					
3								
4	Year	Make	Model	Disp	# Cyl	City	Hwy	Comb
5	2014	Acura	ILX	1.5	4	39	38	38
6	2014	Acura	ILX	2.0	4	24	35	28
7	2014	Acura	ILX	2.4	4	22	31	25
8	2014	Acura	MDX 2WD	3.5	6	20	28	23
9	2014	Acura	MDX 4WD	3.5	6	18	27	21

It finds the first entry that matches – the 1.5 liter engine with 38 MPG. This is a problem if, for example, you want to find the fuel economy of the 2.4 liter sport version. To do that, we need to use **INDEX** and **MATCH**.

Using INDEX and MATCH to Replace VLOOKUP

What we really need is to be able to look up the Model and the Displacement at the same time. **MATCH** is a function that gives you the location of an item in an array. The syntax for **MATCH** is as follows:

```
=MATCH(lookup_value, lookup_array, [match_type])
```

The *lookup_value* is what you are searching for.

The *lookup_array* is the array of values you are trying to find the *lookup_value* in.

The optional *match_type* determines whether **MATCH** must find the *lookup_value* exactly (with a **0**), or return the closest match that comes before it (with a **1**) or after it (with a **-1**) alphanumerically.

The **INDEX** function takes a location and returns the value that is in the cell. The syntax for **INDEX** is as follows:

```
=INDEX(array, row_num, [col_num])
```

The *array* is the table of data that contains the cell value you want.

The *row_num* is the relative row number of the cell you want.

The *col_num* is the relative column number of the cell you want.

By combining **INDEX** and **MATCH** we can produce the same result as **VLOOKUP**. Using the same search we did for **VLOOKUP**, the **INDEX/MATCH** pair looks like this:

	A	B	C	D	E	F	G	H
1	Model	Combined Economy						
2	ILX	=INDEX(C4:H1159,MATCH(A2,C4:C1159,0),6)						
3								
4	Year	Make	Model	Disp	# Cyl	City	Hwy	Comb
5	2014	Acura	ILX	1.5	4	39	38	38
6	2014	Acura	ILX	2.0	4	24	35	28
7	2014	Acura	ILX	2.4	4	22	31	25
8	2014	Acura	MDX 2WD	3.5	6	20	28	23
9	2014	Acura	MDX 4WD	3.5	6	18	27	21
10	2014	Acura	RDX 2WD	3.5	6	20	28	23
11	2014	Acura	RDX 4WD	3.5	6	19	27	22
12	2014	Acura	RLX	3.5	6	20	31	24

```
=INDEX (C4:H1159, MATCH (A2, C4:C1159, 0) , 6)
```

C4:H1159 is the *array* that **INDEX** uses to find the value.

A2 is the cell that the value we want **MATCH** to find.

C4:C1159 is the *lookup_array* that **MATCH** looks through to find the value in **A2**.

The **0** means that **MATCH** will look for the exact value instead of an approximate one.

Column **6** is the column in the **C4:H1159** array that holds the **Combined Fuel Economy** values.

The result is identical to the **VLOOKUP** result. **MATCH** finds the first **Combined Fuel Economy** value for the Acura ILX, which means it will give 38 MPG for the 1.5 instead of one of the other engine options. To find a specific **Model** and engine **Displacement** combination, we need to modify our **INDEX/MATCH** formula into an **array formula**.

Using INDEX and MATCH with Two Criteria

To allow **MATCH** to search for multiple criteria, we are going to change the way it looks for its result by making it an array formula. An array formula takes an array of values instead of a single one and checks each cell in the array until it finds a result.

Our old **MATCH** formula looked like this:

```
=MATCH (A2, C4:C1159, 0)
```

It looked for the value of **A2** in the table **C4:C1159**, and when it found it, it returned the location.

Now we are going to ask it to be creative:

```
=MATCH (1, (C4:C1159=A2) * (D4:D1159=B2) , 0)
```

We have asked **MATCH** to look for a value of 1. Instead of giving it an existing array to look through, we are asking it to build one from scratch. The new array checks all the values in **C4:C1159** for one that matches **A2** and all the values in **D4:D1159** for one that matches **B2**. Where they both match, the array will have a 1 (a **TRUE** boolean result). Where they don't both match, the array will have a 0 (a **FALSE** boolean result). Therefore, **MATCH** will return the location where the array matches 1 (when both of our criteria are true).

If this process doesn't make sense to you, that's okay. Just plug the new **MATCH** function into your **INDEX/MATCH** formula:

```
=INDEX ( C4:H1159, MATCH ( 1, ( C4:C1159=A2 ) * ( D4:D1159=B2 ), 0 ), 6 )
```

When you enter the formula, don't just press **ENTER**. Press **CTRL+SHIFT+ENTER** to tell Excel that it is an array formula. You can tell you've done it right because the entered formula will be surrounded in curly braces {}.

	A	B	C	D	E	F	G	H	I
1	Model	Displacement	Combined Economy						
2	ILX	2.4	25						
3									

With that, your formula will be able to find the Combined Fuel Economy based on both the Model and the Displacement. You can use this technique for any number of criteria with **INDEX** and **MATCH**. Just add additional terms to the multiplication equation.

Multiple Criteria VLOOKUP with INDEX and MATCH Example Download

You can follow along with this tutorial using the original source data and explore an example of the solution in the embedded file below. To download your own copy, click on the green Excel icon in the lower right corner.

	A	B	C	D	E	F	G	H	I
19	2014	Acura	TSX	3.5	6	19	28	23	Naturally Aspirated
20	2014	Acura	TSX WAGON	2.4	4	22	30	25	Naturally Aspirated
21	2014	Aston Martin	DB9	5.9	12	13	19	15	Naturally Aspirated
22	2014	Aston Martin	Rapide S	5.9	12	13	19	15	Naturally Aspirated
23	2014	Aston Martin	V8 Vantage	4.7	8	14	21	16	Naturally Aspirated
24	2014	Aston Martin	V8 Vantage	4.7	8	13	19	15	Naturally Aspirated
25	2014	Aston Martin	V8 Vantage S	4.7	8	14	21	16	Naturally Aspirated
26	2014	Aston Martin	V8 Vantage S	4.7	8	13	19	15	Naturally Aspirated
27	2014	Aston Martin	Vanquish	5.9	12	13	19	15	Naturally Aspirated
28	2014	Audi	A4	2.0	4	24	32	27	Turbocharged
29	2014	Audi	A4 quattro	2.0	4	20	29	24	Turbocharged
30	2014	Audi	A4 quattro	2.0	4	22	32	26	Turbocharged
31	2014	Audi	A5 Cabriolet	2.0	4	24	32	27	Turbocharged
32	2014	Audi	A5 Cabriolet quattro	2.0	4	20	29	24	Turbocharged
33	2014	Audi	A5 quattro	2.0	4	20	29	24	Turbocharged



Andrew Roberts has been solving business problems with Microsoft Excel for over a decade. Excel Tactics is dedicated to helping you master it.

Jump start your analysis with a **one-on-one consultation** or join the newsletter to stay on top of the latest articles. Sign up and you'll get a free guide with 10 time-saving keyboard shortcuts!

[Join Now to Get Your Free Guide!](#)

Other posts in this series...

- [How to VLOOKUP with Multiple Criteria Using INDEX and MATCH](#)
- [Extract a List of Values Filtered by Criteria with Sub-Arrays](#)
- [How to Combine Data from Multiple Rows into One Cell](#)

Share this:

Like 86

Share 2

Tweet 6

More

FILED UNDER TUTORIALS

58 Responses to *How to VLOOKUP with Multiple Criteria Using INDEX and MATCH*

Jason **says:**

February 24, 2014 at 12:13 pm



Thanks for the great post on the use of the combined INDEX & MATCH functions. It will certainly be useful for me in the future. Unfortunately, it falls just short of solving a problem I'm working on now. Can you suggest how this approach might be modified (or another approach entirely) for use with two criteria, one of which is not exact? Specifically, I want to search an array where one value is an exact match, and the second value falls within a range (specified by two columns – an upper and a lower bound). When both of these criteria are met, I want to output the value in a column adjacent to the values that were searched. For example, in the attached image, the cells highlighted yellow are what I'd like to generate a formula to produce (here I determined them by hand). I need to search column A for a match for column F, and then find cases where column G falls between the values in columns B & C. Then I want to output the value contained in column D. Please let me know if you'd like any clarification, and thanks for any help you can provide.

	A	B	C	D	E	F	G	H
1	contig	start postion	end position	ORF		contig	site	ORF
2	1	100	200	A		1	111	A
3	1	300	500	B		1	154	A
4	2	50	600	C		1	307	B
5	2	900	1200	D		1	422	B
6	3	400	500	E		2	62	C
7	3	600	1500	F		2	499	C
8						2	1033	D
9						2	1111	D
10						3	461	E
11						3	474	E
12						3	791	F
13						3	1399	F
14								
15								

Reply

Chuck Trese **says:**

February 27, 2014 at 1:19 pm



Only need to recognize that you can <, as well as =.

For example, in cell H2 enter this formula

=INDEX(\$A\$1:\$D\$7,MATCH(1,(\$A\$1:\$A\$7=\$F2)*(\$B\$1:\$B\$7<\$G2),0),4)

remember to enter using Ctrl-Shift-Enter.

Then you can copy formula into cells H3 through H13

EDITOR'S NOTE: Chuck, I corrected the formula you provided to the way you intended before approving it...

Reply

Chuck Trese says:

February 27, 2014 at 1:24 pm



sorry, text editor won't allow greater-than and less-than symbols.

You just need to recognize that you can use 'greater-than' and 'less-than', not just 'equals' in your comparison checks. (Unfortunately, the equation in the previous post is a mess because it does not show the greater-than or less-than symbols, but instead removed the middle of the formula – the part that mattered the most.)

Reply

Andrew Roberts says:

February 27, 2014 at 1:35 pm



Hi Jason,

Chuck's suggestion to use a greater-than or less-than to set upper and lower bounds is right on the money. His formula checks to make sure Column B is less than the Column G item. To be more precise, you could check the upper bound too, like so:

```
=INDEX($A$1:$D$7,MATCH(1,($A$1:$A$7=$F2)*($B$1:$B$7< $G2)*($C$1:$C$7>$G2,0),4)
```

Don't forget to enter as an array formula with CTRL+SHIFT+ENTER!

Reply

Randy says:

February 28, 2014 at 2:26 pm



Thanks for the fantastic post! I have one question. Sometimes when I am using a two criteria formula with index and match just as you described. I want to return a "0" or blank if the result is #N/A.

How do I change the formula to return 0 or something else when both criteria are not met?

Reply

Andrew Roberts says:

February 28, 2014 at 4:02 pm



Hi Randy,

Glad you like the tutorial! If you want to catch errors like #N/A, you need to wrap the INDEX(MATCH()) in an IFERROR() function. This lets you specify the behavior when you have an error. For more details, check out this [guide on Excel error handling](#).

Reply

Kelly says:

March 26, 2014 at 5:07 am



I'm trying to add the IFERROR to an index/match function and can't seem to get it right. Can you be specific about how to add this?

Reply

Andrew Roberts says:

March 26, 2014 at 12:16 pm



Hi Kelly,

You might find my guide on error handling useful... Specifically, the section on IFERROR and Common Error Handling Techniques. You can find it [here](#).

Reply

Thijs says:

March 5, 2014 at 6:54 pm



Great post, and almost fixes my problem, I'm just wondering how you can use this if you have multiple

similar instances of the searched criteria (say two people with the same first and last name) and therefore multiple outcomes exist. Is there a way to get excel to look up the 1st, 2nd, 3rd etc. of each combined instance so I can put them in a row. I've attached an example of what I'm trying to do. And thanks for any help you can provide

	A	B	C	D	E	F	G	H	I	J	K	L	M	N	O	P
1		Input														
2	Population	Species	Well ID	POP	Spec											
3	Mt Lewis	Synoum paniculosum	P6:8E	Mt Lewis	Synoum paniculosum	P6:8E	P6:8F	P6:8G	P6:8H	P6:9A	P6:9B	P6:9C	P6:9D	P6:9E	P6:9F	P6:8B
4	Mt Lewis	Synoum paniculosum	P6:8F	Mt Lewis	Niemeyera prunifera	P6:2D	P6:2E	P6:2F	P6:2G							
5	Mt Lewis	Synoum paniculosum	P6:8G													
6	Mt Lewis	Synoum paniculosum	P6:8H													
7	Mt Lewis	Synoum paniculosum	P6:9A													
8	Mt Lewis	Synoum paniculosum	P6:9B													
9	Mt Lewis	Synoum paniculosum	P6:9C													
10	Mt Lewis	Synoum paniculosum	P6:9D													
11	Mt Lewis	Synoum paniculosum	P6:9E													
12	Mt Lewis	Synoum paniculosum	P6:9F													
13	Mt Lewis	Niemeyera prunifera	P6:2D													
14	Mt Lewis	Niemeyera prunifera	P6:2E													
15	Mt Lewis	Niemeyera prunifera	P6:2F													
16	Mt Lewis	Niemeyera prunifera	P6:2G													
17	Mt Lewis	Synoum paniculosum	P6:8B													
18																

Reply

Andrew Roberts says:

March 6, 2014 at 12:32 am



Hi Thijs,

Depending on what you're trying to do, you might find [How to Combine Data from Multiple Rows into One Cell](#) or [Extract a List of Values Filtered by Criteria with Sub-Arrays](#) useful.

Reply

Thijs says:

March 12, 2014 at 1:20 pm



Hey Andrew,

Thanks for the reply, I actually figured it out on my own, I just used Vlookup but simply combined the two search criteria and the two rows where to find it using an & function That way I could simply ask it to find outcome 1 to 15 and get my results.

Reply

Devin says:

March 26, 2014 at 8:54 am



Thanks, helped alot. Although I have a table of around 7000 and this method takes alot of time. What would be a faster way to do this. I think it is slow because it has to create an array for each cell. It does about 5.8 every sec. = 20 mins of waiting. Just in case it is relevant, no other calculations are on the sheet, and I am using 3 critiera.

Reply

Andrew Roberts says:

March 26, 2014 at 12:30 pm



Hi Devin,

Because the INDEX/MATCH has to compare each criteria against the entire table, lots of rows and lots of criteria makes the number of necessary calculations grow exponentially. Try to think of ways to pre-calculate some of the criteria to reduce the total number in the INDEX/MATCH or search a sub-set of the data where possible. Also, if you are working with the table often, it may be worth it to shut off automatic re-calculation and only re-run the 20 minute session when you are finished making changes.

Good luck!

Andrew

Reply

chris says:

March 26, 2014 at 3:25 pm



I'm looking for a formula that look a 3 criteria, item, size and month..I wrote the formula many different ways but can't seem to get it right can you pls advise. thanks ahead for your help

Brand	Item	Size	Mon	Dollar Volun
pepsi	Pepsi 20pk	2L	1	\$230.04
pepsi	Pepsi 20pk	1L	10	\$94.50
pepsi	pepsi 6pk	12oz	11	\$134.76
pepsi	Diet Pepsi	750ml	3	\$129.36
pepsi	cherry pepsi	750ml	1	\$129.96
pepsi	cherry pepsi	750ml	8	\$138.48
pepsi	Diet cherry pepsi	1.75L	3	\$107.24
pepsi	Diet cherry pepsi	1.75L	8	\$112.44
pepsi	Diet cherry pepsi	1.75L	10	\$112.44

Reply

Andrew Roberts says:

March 26, 2014 at 3:33 pm



Hi Chris,

The basic format for a 3-criteria INDEX/MATCH would be as follows:

=INDEX(TableRange,MATCH(1,(CriteriaRange1=Criteria1)*(CriteriaRange2=Criteria2)*(CriteriaRange3=Criteria3),0),ReturnColumnNumber)

If you have more specific difficulties, I'd suggest you submit a question at one of the forums available on my [Web Resources](#) page.

Good luck!

Andrew

Reply

Rachel says:

March 31, 2014 at 11:41 am



Hi Andrew,

I have the same issue as Chris does (matching 3 criteria), plus I need to match a few columns of data, and I need to pull it from one tab into another tab. Is that possible?

Rachel

Reply

Rachel says:

April 1, 2014 at 11:50 am



Thanks Andrew! If this works, it will save myself and my team hours of pressured manual excel labor.

Reply

Ken says:

May 16, 2014 at 12:55 am



Thanks for taking the time to write this fantastic explanation for using INDEX and MATCH for multiple criteria. I've actually been using the MATCH part for a year or so, but I had no idea what the 1 aspect of it was about. Yours is the first article that explained the logic behind it. I never comment on blog posts, but I had to for this – it was excellent!

Much appreciated!

Reply

Jonas says:

May 20, 2014 at 2:05 am



Thanks for this great tutorial. I finally understand how to use the Index + Match function and was able to use it in a complex spreadsheet for work. In the end, I had to transpose the data from a vertically structured table to a horizontally structured one. Is there a way to use Index + Match for data that is structured from top to bottom? Either way, a great tutorial and well written blog. Keep up the good work!

Reply

James says:

May 20, 2014 at 9:35 am



Thanks! Great Stuff!

How many criteria can I fit in using this method?

Best regards!

Reply

Andrew Roberts says:

May 20, 2014 at 3:44 pm



Hi Jonas –

You can use INDEX/MATCH vertically or horizontally... You just need to use the MATCH statement in either the row number or column number field. Syntax for INDEX is as follows: =INDEX(array, row_num, [col_num])

Reply

Andrew Roberts says:

May 20, 2014 at 4:43 pm



Hi James –

As many criteria as you want... I warn, however, that each additional criteria adds computation time. If you do more than two, things start slowing down exponentially. It may be easier to concatenate multiple criteria into one to save on computation.

Reply

Brandon says:

May 23, 2014 at 11:53 am



Thank you so much for the step by step instructions. I can't tell you how much time this saved me!

Reply

Takis says:

May 26, 2014 at 6:48 am



Thank you Andrew! Very nice guide and extremely useful!

"If this process doesn't make sense to you, that's okay. Just plug the new MATCH function into your INDEX/MATCH formula:",

hahahahaha! really enjoyed when I read it!

Reply

Dustin says:

May 28, 2014 at 4:09 am



This almost does exactly what i'm looking to do. I just want to know is it possible to generate more results. so If i wanted to generate a list from my table of everyone with a certain qualification.

Reply

Andrew Roberts says:

May 28, 2014 at 3:01 pm



Hi Dustin,

I think you might find this tutorial useful:

<http://exceltactics.com/make-filtered-list-sub-arrays-excel-using-small/>

Reply

Katie says:

June 3, 2014 at 6:19 am



Thank you – this is a big help. I was able to do the formula with 3 variables. My only issue is I'd like to copy

this formula down an entire column of 100K+ line items. I am unable to just click the bottom right corner because I get a #REF! error. The formula copies with “#REF!” where the match range is supposed to be. The only way I can get it to work is hit the CTRL+SHIFT+ENTER in each cell, but I cannot do this 100K+ times. Any simple way to copy this formula down an entire column?

Reply

Andrew Roberts says:

June 3, 2014 at 10:17 am



Hi Katie,

Congratulations on getting the formula up and running! Array formulas (the ones you enter with CTRL+SHIFT+ENTER) are a bit picky about how you copy them. You can select the cell with the entered formula and drag down with the bottom right corner to copy the formula down. You can also copy and paste the formula into other cells, but you must exclude the original cell from the paste selection or Excel will give an error.

As an aside, I will caution you that a 3-criteria INDEX/MATCH applied across 100k cells will likely take a very long time to compute (on the order of *hours*). Be sure to save your work before copying the formula down and plan to do so when you are able to leave Excel alone to process for several hours.

Good luck!

Andrew

Reply

alberto Jaume says:

June 5, 2014 at 9:12 am



Hi Andrew,

This is a very nice article/tutorial! I have found it can be used in different circumstances, however I was wondering whether you can provide further explanation in regards to a particular situation. I am trying to produce a report that checks for a name in table (that is repeated multiple times) and selects the most recent entry (each entry has a different date in a different column). I am breaking my head trying to use your approach but not sure how it could make it work. Could you cast some light on this?

Many thanks!

Reply

Andrew Roberts says:

June 6, 2014 at 11:31 am



Hi Alberto,

Thanks for the feedback! This method should always return the first match it finds, so if you pre-sort your list so that newer entries appear at the top, the lookup should work the way you want.

Good luck!

A

Reply

Unal says:

June 10, 2014 at 4:08 am



Hi,

I think there is an easy way to do it (but you need a new column in your data sheet)

Please check below.

I merged the cells that I want to use in criteria and put them in column C together (=A6&”&B6)

Then used this formula to match it. =VLOOKUP(G6&”&H6,\$C\$6:\$D\$11,2,FALSE)

That's it. You can even create as much criteria as you want.

Regards

	A	B	C	D	E	F	G	H	I	J	K
1											
2											
3											
4			you can hide								
5	Item	Colour	New column	Price £			Item	Colour	Sold	Price	Value
6	Box	Black	Box Black	34.00			Box	Black	100	=VLOOKUP(G6 & " "&H6,\$C\$6:\$D\$11,2,FALSE)	3,400
7	Box	White	Box White	36.00			Box	White	100	36.00	3,600
8	Paper	Green	Paper Green	36.00			Box	Black	200	34.00	6,800
9	Paper	Blue	Paper Blue	44.00			Box	Black	250	34.00	8,500
10	Pencil	Black	Pencil Black	45.00			Paper	green	300	36.00	10,800
11	Pencil	Red	Pencil Red	12.00			Paper	green	100	36.00	3,600
12							Pencil	black	100	45.00	4,500
13							Pencil	black	50	45.00	2,250
14							Pencil	red	65	12.00	780
15							Box	Black	50	34.00	1,700
16							Box	Black	65	34.00	2,210
17							Paper	green	200	36.00	7,200
18							Paper	green	250	36.00	9,000
19							Pencil	black	300	45.00	13,500
20							Pencil	red	100	12.00	1,200
21							Box	White	100	36.00	3,600
22											

Reply

Alberto **says:**

June 12, 2014 at 5:34 am



This How to saved me about 5 days woth of work... its is awesome. Thanks a bunch.

Reply

Venkatesh **says:**

June 14, 2014 at 12:37 am



Really very helpful, thanks lot.

Reply

chris **says:**

June 16, 2014 at 4:23 am



When I download your example in Excel 2007, it doesn't return an answer in cell C2 but rather #VALUE!

Any ideas why? I didn't change anything.

Reply

Andrew Roberts **says:**

June 16, 2014 at 11:10 pm



The formula in C2 of the example worksheet is an array formula. These were available as early as Excel 2003. If there is a #VALUE! error in cell C2, the formula has probably been re-entered by pressing only ENTER instead of CTRL+SHIFT+ENTER. If you can't see the {} curly brackets around the =INDEX() command in the formula box, you need to click in the formula box and re-enter the array formula using CTRL+SHIFT+ENTER.

Reply

chris **says:**

June 17, 2014 at 2:44 am



Great that works!

Reply

Zane **says:**

June 18, 2014 at 3:27 am



Reply to: Unal comment below. Thank you very much. This formula has saved my life. This is genius thanks again!!!!

Hi,

I think there is an easy way to do it (but you need a new column in your data sheet)

Please check below.

i merged the cells that I want to use in criteria and put them in column C together (=A6&" "&B6)

Then used this formula to match it. =VLOOKUP(G6&" "&H6,\$C\$6:\$D\$11,2,FALSE)

That's it. You can even create as much criteria as you want.

Regards

Reply

Dwight says:

June 20, 2014 at 5:49 pm



This is very cool. Thanks for sharing. I had good luck with ctrl R and ctrl D to copy down and right. I didn't know about IFERROR. I routinely use =IF(ISERROR(formula),"",formula) to get blanks instead of #N/A or #VALUE. Cheers.

Reply

Vince says:

June 21, 2014 at 9:09 am



Thanks very much for sharing your knowledge. This worked perfectly for me!

Reply

Sbu Nkosi says:

June 29, 2014 at 8:48 am



I have a different (and maybe easier) challenge. I'm trying to do the sum of all the values that meet a criteria i specify. For example, I have a 2-column table/array that consists of months and their values. I have several rows containing Jan 2015 and want to add up all values for Jan 2015. How do i do this?

Reply

Andrew Roberts says:

July 7, 2014 at 1:54 pm



Sounds like a job for SUMIF() or SUMIFS() to me...

Reply

Sean says:

July 11, 2014 at 12:58 pm



Hello,

I am having a problem getting the formula to work I keep getting a result of N/A even though I know there is a match.

Here is my formula

=INDEX(Sheet1!\$B\$6:\$AD\$32,MATCH(1,(Sheet1!\$B\$6:\$AD\$32=Sheet2!\$A5)*
(Sheet1!\$C\$6:\$AD\$32=Sheet2!H\$2),0),Sheet3!\$D\$3)

Reply

Andrew Roberts says:

July 20, 2014 at 6:28 pm



Check to make sure the formula is entered with CTRL+ALT+ENTER and not just ENTER. You'll see curly braces {} around the formula if it is done correctly.

Reply

Juliana says:

August 6, 2014 at 1:09 pm



Hi Andrew, I'm having a similar issue. When I drag the formula into the rest of the column, it changes the array I'm searching through. I want it to search through. In my first cell the formula is {=INDEX(Sheet2!A1:C61,MATCH(1,(Sheet2!A1:A61=DATA!C2)*(Sheet2!B1:B61=DATA!D2),0),3)} but when I drag into the next cell it changes the formula to {=INDEX(Sheet2!A2:C62,MATCH(1,(Sheet2!A2:A62=DATA!C3)*(Sheet2!B2:B62=DATA!D3),0),3)}

How can I keep that fixed?

E3					=INDEX(Sheet2!A2:C62,MATCH(1,(Sheet2!A2:A62=DATA!C3)*(Sheet2!B2:B62=DATA!D3),0),3)}
1	Workqueue ID	Workqueue Name	Academic Department	Specialty	Column1
2	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	Asmin Santana
3	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	#N/A
4	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	#N/A
5	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	#N/A
6	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	#N/A
7	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
8	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
9	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
10	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
11	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
12	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
13	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
14	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
15	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
16	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
17	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
18	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
19	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
20	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	
21	3368	WCMC POBO ENROLLMENT	Anesthesiology	Anesthesiology	

Reply

Andrew Roberts says:

August 8, 2014 at 1:41 pm



All you need to do is lock the row references. Sheet2!A1:A61 becomes Sheet2!\$A\$1:\$A\$61.

Reply

Arif says:

August 12, 2014 at 7:54 am



Thanks Andrew, the formula is working fine, there is a small issue, when Match does not find the value its giving me #N/A, will you please tell me how I can use IFERROR,, ISNA or some other function to avoid this issue.

`=INDEX(Order!A2:C1000,MATCH(1,(Order!A2:A1000=K$1)*(Order!$B$2:$B$1000=Production!$A3),0),3)}`

Reply

Andrew Roberts says:

August 12, 2014 at 10:45 am



Hi Arif,

The **IFERROR()** formula is pretty straightforward. Just use the form **IFERROR(original_formula, formula_if_error)**. You'll still need to enter with **CTRL+SHIFT+ENTER**. For more error handling options, you might like the [Definitive Guide to Excel Error Types and Error Handling](#).

Reply

Wilder says:

July 15, 2014 at 2:58 pm



Hi Andrew,

Thanks for this post. Please, letting me to know if it's possible to avoid the array option. It works for one cell (one record) at the time. I wanna use a formula for many records and drag the formula to complete the search. Much obliged,

Reply

Andrew Roberts says:

July 20, 2014 at 6:11 pm



You should still be able to drag down the formula after you enter it as an array. Another option is to concatenate all your criteria into a single string to avoid the "multiple criteria" INDEX/MATCH all together...

Reply

Wilder says:

July 28, 2014 at 12:53 pm



Thanks Andrew. Can you please show me an example for the single string option? Thanks a lot!

Reply

nsdkljvbdvib says:

July 31, 2014 at 12:53 pm



How'd you get away with the implied if/then? In my own implementation I have to add IF() functions to get the Boolean values.

Oddly, your example works when I download it. The only operational difference between yours and mine is that I'm referencing values in a separate workbook.

Reply

Adam says:

August 5, 2014 at 9:34 am



Hi,

Thanks for sharing this very useful formula. I am using this to reference data from two separate Tables. I attempted to use the Table name and Table Columns as reference in the Match formula. It looked like this:

```
{=Match(1,(Inventory[Page ID]=[@Page])*(Inventory[Language]=[@Language]),0)}
```

However, this did not work. I had to use absolute cell references – e.g., A2:A100. Any idea why this would be? Perhaps I am just referencing incorrectly?

Cheers,

Adam

Reply

Andrew Roberts says:

August 8, 2014 at 1:49 pm



Hi Adam,

Data Tables in Excel work on internal logic that lets you easily duplicate formulas all the way down the rows. Your reference @Page is referring to a relative page in the row of the table. It is likely not interpreting this correctly for the MATCH, which generally works on absolute row references. There may be a way to use table references in INDEX/MATCH calls, but I have not explored the topic myself. Sounds like you got it to work, though!

Andrew

Reply

Phil Mullins says:

August 14, 2014 at 1:25 pm



Thanks for this great tutorial – it is exactly what I needed, and is the most well-written and easy to follow post of its kind that I've seen. Keep up the good work!

Reply

RAJ says:

August 17, 2014 at 10:31 pm



Very useful details for cell comparison

Reply

R Kaur says:

August 20, 2014 at 8:00 am



Thanks a lot for posting this.

It really helped me solve a problem that I was struggling with.

Reply

Lori H says:

August 21, 2014 at 12:43 pm



This is a great article but I have a slightly different issue. I have two sheets of data that have three of the same columns that I want to compare the row fields. If there is not a match from one sheet to another then I need to generate a report of those differences or at least highlight the non-matches. Any ideas on how to do that? Thanks!

Reply

Andrew Roberts says:

August 21, 2014 at 12:45 pm



You may find my article on Sub-Arrays useful: [Extract a List of Values Filtered by Criteria with Sub-Arrays](#)

Reply

Leave a Reply

Your email address will not be published. Required fields are marked *

Name *

Email *

Website

Comment

POST COMMENT

Select an image for your comment (GIF, PNG, JPG, JPEG):

Choose File

No file chosen

☐

Notify me of follow-up comments by email.

☐

Notify me of new posts by email.